11-28-95

SANDIA REPORT

SAND95–1697 • UC–910
Unlimited Release
Printed November 1995

Diffraction Efficiency Analysis for Multi-Level Diffractive Optical Elements

Ireena A. Erteza

Prepared by
Sandia National Laboratories
Albuquerque, New Mexico 87185 and Livermore, California 94550
for the United States Department of Energy
under Contract DE-AC04-94AL85000



SF2900Q(8-81)

Issued by Sandia National Laboratories, operated for the United States Department of Energy by Sandia Corporation.

Printed in the United States of America. This report has been reproduced directly from the best available copy.

Available to DOE and DOE contractors from
Office of Scientific and Technical Information
PO Box 62
Oak Ridge, TN 37831

Prices available from (615) 576-8401, FTS 626-8401

Available to the public from
National Technical Information Service
US Department of Commerce
5285 Port Royal Rd
Springfield, VA 22161

NTIS price codes
Printed copy: A03
Microfiche copy: A01

SAND95-1697
Unlimited Release
Printed November 1995

Distribution
Category UC-910

Diffraction Efficiency Analysis for Multi-Level Diffractive Optical Elements

Ireena A. Erteza
Optical Systems and Image Processing Department
Sandia National Laboratories
Albuquerque, NM 87185

Abstract

Passive optical components can be broken down into two main groups: refractive elements and diffractive elements. With recent advances in manufacturing technologies, diffractive optical elements are becoming increasingly more prevalent in optical systems. It is therefore important to be able to understand and model the behavior of these elements.

In this report, we present a thorough analysis of a completely general diffractive optical element (DOE). The main goal of the analysis is to understand the diffraction efficiency and power distribution of the various modes affected by the DOE. This is critical to understanding cross talk and power issues when these elements are used in actual systems.

As mentioned, the model is based on a completely general scenario for a DOE. This allows the user to specify the details to model a wide variety of diffractive elements. The analysis is implemented straightforwardly in Mathematica. [7] This report includes the development of the analysis, the Mathematica implementation of the model and several examples using the Mathematical analysis tool. It is intended that this tool be a building block for more specialized analyses.

Contents

List of Figures

1 Introduction

This analysis addresses the efficiency and power distribution between different modes in a diffractive optical element (DOE). This is useful in determining cross-talk issues among modes.

In this analysis, effort has been made to make the analysis as general as possible. In doing so, there are many parameters available to the user to describe the exact situation to be analyzed.

We will begin by describing the general scenario. In the following section, the actual theory and equations used will be derived and presented. Then the implementation in Mathematica will be discussed, and some example calculations shown.

1.1 General Scenario

The general scenario is shown in Figure 1. Light is incident at angle θ in medium of index μ_R. It hits a diffractive optical element of medium of index μ. Because of the change in index of refraction, the light inside the DOE has an angle θ'. The thickness of the n^{th} step in the diffractive optical element is d_n. The medium underneath the DOE has index μ_L.



Figure 1: General Scenario for DOE Analysis

From the analysis, we want to know if light is incident on the DOE, how much power goes into the various diffracted orders, and what is the power distribution for those different orders at various distances. Then, the power hitting the receiver or other locations can be calculated.

1.2 Diffractive Optical Elements

A diffractive lens emulates the phase function of a refractive lens, using a phase grating. The phase grating imparts the same phase delay as a refractive lens, but modulo 2π. The modulo 2π grating is called a blaze.

If the phase function has continuous phase levels, then 100% efficiency can be obtained. Typically, however, the continuous valued phase function is approximated by discrete values, giving rise to "Binary Diffractive Optics". The analysis to be presented is directed toward binary diffractive optical elements.

The phase function of a diffractive lens is periodic in $\frac{r^2}{2f}$, with period λ. Therefore, the lens transmittance can be expanded into a Fourier series in the variable $\frac{r^2}{2f}$ or $\frac{r^2}{2f\lambda}$. The blaze as a function of $\frac{r^2}{2f}$ or $\frac{r^2}{2f\lambda}$ is a stepped ramp. Fourier decomposition describes this stepped ramp as the weighted sum of terms $\exp(-i2\pi qx)$, where $x = \frac{r^2}{2f\lambda}$, and q is an integer. Recall that the phase of a spherical wave is $\exp(-i\frac{2\pi}{\lambda}\frac{r^2}{2f})$ for a wave with focal length f. Therefore, the various orders have focus at $\frac{f}{q}$.

If the medium has index η, then a plane wave incident on a phase plate with phase $\exp(-iq\frac{2\pi}{\lambda}\frac{\eta}{\eta}\frac{r^2}{2f})$ results in a wave identical to a converging spherical wave in media η, converging at a focus, $\frac{\eta f}{q}$.

To summarize, the focal power of a lens depends *only* on what kind of phase delay is imparted to the wave exiting the lens. For a refractive lens, the phase delay distribution will be of the form: $\psi(r) = k_0(f - \sqrt{f^2 + r^2})$. An incident plane wave then becomes $\exp(i(\omega t - \psi(r))$. If $f \gg r$, then $\psi(r) \approx \frac{-\pi r^2}{\lambda f}$.

The Fourier analysis of a diffractive lens is based on the following statements:

1. The lens is a phase plate.
2. Taking the Fourier series of the phase decomposes an incoming plane wave into a series of converging and diverging spherical waves.
3. The focal lengths of these waves is dependent on the periodicity of the phase plate.

2 Theory

2.1 Normal Incidence

Consider the problem of a plane wave illuminating the grating shown in Figure 2.

2.1.1 Phase Function of Grating

The effect of the grating is to impart a phase function onto the incoming light. For an incident plane wave, the phase function after passing through the grating can be given by $\exp[-i\psi(\xi, \lambda)]$.



Figure 2: Diffractive Grating

If surrounding medium is air, and

$$P = \text{number of step levels in the grating period}$$

$$\Delta = \text{one grating period}$$

$$0, d_1, 2d_1, 3d_1 \ldots (P-1)d_1 = \text{geometrical heights of the steps}$$

$$d_n = nd_1 \qquad \text{for } n = 0,1,2\ldots P-1, \text{ or } \frac{n}{P}\Delta \le \xi \le \frac{n+1}{P}\Delta$$

$$\mu(\lambda) = \text{index of refraction of step material}$$

$$\theta = \text{angle of incidence}$$

then

$$\psi_n(\lambda) = n\psi_1(\lambda) \tag{1}$$

The phase steps $\psi_n(\lambda)$ form a stepped phase function $\psi(\xi, \lambda)$

$$\psi(\xi, \lambda) = n\psi_1(\lambda), \qquad \frac{n}{P}\Delta \le \xi \le \frac{n+1}{P}\Delta, \tag{2}$$

$n = 0, 1, 2 \cdots P-1$.

$$\psi_1(\lambda) = 2\pi[\mu(\lambda) - 1]\frac{d_1}{\lambda} \tag{3}$$

For maximum efficiency on a blaze, $\psi_n = \frac{n}{P}2\pi$, $0 \le n \le P-1$, resulting in

$$d_1 = \frac{\lambda_0}{(\mu - 1)P}, \tag{4}$$

where $\mu = \mu(\lambda_0)$ and λ_0 is the design wavelength.

2.1.2 Diffraction Efficiency

If a grating is illuminated by a plane wave, the light is split into several diffraction orders. The power into the different orders can be found by taking the square of the coefficients of the Fourier series.

Let $a_q(\lambda)$ be the fraction of radiant energy at wavelength λ diffracted into order q by the grating.

$$a_q(\lambda) = \left| \frac{1}{\Delta} \int_0^{\Delta} \exp[-i\psi(\xi, \lambda)] \exp[2\pi i q \frac{\xi}{\Delta}] d\xi \right|^2 \tag{5}$$

$$a_q(\lambda) = \left| \sum_{n=0}^{P-1} \int_{\frac{n}{P}}^{\frac{n+1}{P}} \exp[-in\psi_1(\lambda)] \exp[2\pi i q x] dx \right|^2 , \tag{6}$$

where $\xi = \Delta \cdot x$. Then using $\sum_{m=0}^{N-1} \exp\left[\frac{-i2\pi}{N} Am\right] = \frac{\sin[\pi A]}{\sin[\pi A/N]}$,

$$a_q(\lambda) = \left[\frac{\sin(\pi \frac{q}{P})}{\pi \frac{q}{P}} \right]^2 \left[\frac{\sin[P(\frac{1}{2}\psi_1(\lambda) - \pi \frac{q}{P})]}{P \sin[\frac{1}{2}\psi_1(\lambda) - \pi \frac{q}{P}]} \right]^2 \tag{7}$$

2.2 Non-Normal Incidence

A more general scenario includes the possibility of the light coming in at an angle, as in Figure 1. For non-normal incidence, the phase delay imparted by the grating must be calculated as a function of the incident angle. In addition, the phase function of the incoming plane wave is no longer constant along the ξ direction; it is a function of ξ.

2.2.1 Phase Function of Grating

The total phase delay *for normal incidence* is given by:

$$\phi = k\mu d_n + k\mu_R(d_p - d_n)$$

$$\phi = k\mu_R d_p + k(\mu - \mu_R) d_n,$$

where $k = 2\pi/\lambda$ and d_p is the height of the thickest step.

From Snell's law, if the incident angle is θ, $\mu_R \sin(\theta) = \mu \sin(\theta')$. The total phase delay is then:

$$\phi = k\mu \frac{d_n}{\cos\theta'} + k\mu_R \frac{(d_p - d_n)}{\cos\theta}$$

$$\phi = k\mu_R \frac{d_p}{\cos\theta} + k(\frac{\mu}{\cos\theta'} - \frac{\mu_R}{\cos\theta}) d_n,$$

So the phase delay in going through the n^{th} step is

$$\psi_n = \frac{2\pi}{\lambda} \frac{1}{\cos\theta' \cos\theta} (\mu\cos\theta - \mu_R \cos\theta') d_n \tag{8}$$

If the DOE is designed for maximum efficiency for non-normal incidence, then

$$\psi_n = \frac{n2\pi}{P} = \frac{2\pi}{\lambda} \frac{1}{\cos\theta' \cos\theta} (\mu\cos\theta - \mu_R \cos\theta') d_n, \tag{9}$$

resulting in

$$d_n = \frac{n\lambda_0}{P} \frac{\cos\theta' \cos\theta}{(\mu\cos\theta - \mu_R \cos\theta')}, \tag{10}$$

where $d_1 = d_n/n$.

2.2.2 Diffraction Efficiency

If the light incident on the grating is not normal to the grating, then the phase of the wave is not constant over the grating. The phase after the grating is the product of the phase incident on the grating and the transmittance function of the grating. To find the diffraction efficiency of the various orders of the grating, we need to find the Fourier decomposition of the product of the incoming phase and the phase function of the grating.

To describe the incoming phase distribution from a non-normal plane wave, consider Figure 3. If λ is the wavelength of the illumination, then the phase function of the plane wave is given by

$$\exp\left[\frac{-i2\pi\mu_R}{\lambda}(\xi\sin\theta)\right].$$



Figure 3: Plane Wave Incident at Non-Normal Incidence

The fraction of radiant energy at wavelength λ diffracted into order q by the grating, $a_q(\lambda)$, is then

$$a_q(\lambda) = \left|\frac{1}{\Delta}\int_0^{\Delta} \exp\left[\frac{-i2\pi\mu_R}{\lambda}(\xi\sin\theta)\right] \exp[-i\psi(\xi,\lambda)] \exp\left[2\pi i q\frac{\xi}{\Delta}\right] d\xi\right|^2, \tag{11}$$

where

$$\xi = \Delta \cdot x$$

and

$$\psi(\xi,\lambda) = n\psi_1(\lambda), \qquad \frac{n}{P}\Delta \leq \xi \leq \frac{n+1}{P}\Delta,$$

$n = 0, 1, 2 \cdots P-1$.

With the help of Figure 4, we can rewrite this as

$$a_q(\lambda) = |I|^2 \tag{12}$$

$$I = \int_0^1 \exp\left[\frac{-i2\pi}{\lambda}(\Delta \cdot x)\mu_R \sin\theta\right] \exp[-i\psi(x,\lambda)] \exp[2\pi i q x] dx \tag{13}$$

$$\psi(x,\lambda) = n\psi_1(\lambda), \qquad \frac{n}{P} \le x \le \frac{n+1}{P}. \tag{14}$$

$$I = \sum_{n=0}^{P-1} \int_{\frac{n}{P}}^{\frac{n+1}{P}} \exp\left[\frac{-i2\pi}{\lambda}(\Delta \cdot x)\mu_R \sin\theta\right] \exp[-in\psi_1(\lambda)] \exp[2\pi i q x] dx \tag{15}$$



Figure 4: Grating Geometry with Various Variables

First, let's do the integral, integrating only those things that are a function of x.

$$\int_{\frac{n}{P}}^{\frac{n+1}{P}} \exp\left[\frac{-i2\pi}{\lambda}(\Delta \cdot x)\mu_R \sin\theta\right] \exp[2\pi i q x] dx =$$

$$\frac{1}{\pi(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)} \left[\exp\left[i\frac{\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)\right]\right] \cdot$$

$$\exp\left[i\frac{2\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)n\right] \sin\left[\frac{\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)\right] \tag{16}$$

Substituting this result in the equation for I,

$$I = \frac{1}{\pi(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)} \exp\left[i\frac{\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)\right] \sin\left[\frac{\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)\right] \cdot$$

$$\sum_{n=0}^{P-1} \exp\left[i\frac{2\pi}{P}(q - \frac{\Delta}{\lambda}\mu_R \sin\theta)n\right] \exp[-in\psi_1] \tag{17}$$

Looking at just the summation and recalling

$$\sum_{m=0}^{N-1} \exp\left[\frac{-i2\pi}{N}Am\right] = \frac{\sin[\pi A]}{\sin[\pi A/N]}$$

yields

$$\sum_{n=0}^{P-1} \exp\left[\frac{-i2\pi n}{P}\left(-(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)+\frac{\psi_1 P}{2\pi}\right)\right] = \frac{\sin\left[P\left(\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right)\right]}{\sin\left[\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right]}. \quad (18)$$

Therefore we have:

$$I = \exp\left[i\frac{\pi}{P}(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\right]\frac{\sin\left[\frac{\pi}{P}(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\right]}{\left[\frac{\pi}{P}(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\right]}\frac{\sin\left[P\left(\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right)\right]}{P\sin\left[\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right]}. \quad (19)$$

$a_q(\lambda) = |I|^2 = II^*$, therefore

$$a_q(\lambda) = \left[\frac{\sin\left[\frac{\pi}{P}(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\right]}{\left[\frac{\pi}{P}(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\right]}\right]^2\left[\frac{\sin\left[P\left(\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right)\right]}{P\sin\left[\frac{\psi_1}{2}-\frac{(q-\frac{\Delta}{\lambda}\mu_R\sin\theta)\pi}{P}\right]}\right]^2, \quad (20)$$

where $\psi_1 = \psi_1(\lambda)$.

2.2.3 Focal Lengths for Various Orders

The actual focal length of a diffractive lens depends only on the periodicity of the phase function it imparts, and on the medium after the lens.

Recall that a blazed grating is periodic in the quantity $\frac{r^2}{2F\Lambda}$ with period 1. This means that if light of wavelength Λ is incident on this element and then passes through air, that the first order focus will be at F. We use the symbols Λ and F to emphasize that these are "design" parameters.

If wavelength of light λ is incident on a grating that is periodic in $\frac{r^2}{2F\Lambda}$ with period 1, and then passes through air, the first order focus will be at $\frac{F\Lambda}{\lambda}$. This is because from the Fourier decomposition, we have a phase function broken into components $\exp[-i2\pi qx]$. If this wave is propagating in air, then

$$\exp[-i2\pi qx]\,|_{x=\frac{r^2}{2\Lambda F}} = \exp\left[\frac{-i2\pi q}{\lambda}\frac{r^2}{2F\Lambda/\lambda}\right].$$

If this wave is propagating in a medium with index η, then

$$\exp[-i2\pi qx]\,|_{x=\frac{r^2}{2\Lambda F}} = \exp\left[\frac{-i2\pi\eta q}{\lambda}\frac{r^2}{2F\eta\Lambda/\lambda}\right],$$

meaning that the first order focus in medium of index η is $\frac{F\eta\Lambda}{\lambda}$.

Note, for a DOE, the only fixed physical parameter is the groove spacing. If you take a DOE, you will find that it is periodic in r^2. Let that measured period be designated T_m. For light of wavelength λ_i incident on this DOE, the first order focus will be at $\frac{T_m}{2\lambda_i}$.

2.3 Uniform Distribution Power Calculations

The paraxial focus of the q^{th} diffracted order is at a distance $\frac{f}{q}$ away from the lens. Using geometrical optics, one can calculate the radius, x'_q, at plane z where the extreme ray from the edge of the zone plate which passes through the q^{th} order focus strikes. See Figure 5. This quantity, x'_q, is only meaningful for distances greater than 0. ($z = 0 \Rightarrow$ the plane of the DOE.)



Figure 5: Geometrical Picture of Extreme Ray of qth Order

$$
\begin{aligned}
x'_q &= w_0, & q &= 0 & & \\
x'_q &= w_0(1 - \frac{zq}{f}), & q &> 0 & z &< \frac{f}{q}. \\
x'_q &= w_0(\frac{zq}{f} - 1), & q &> 0 & z &\geq \frac{f}{q} \\
x'_q &= w_0(1 - \frac{zq}{f}), & q &< 0 & z &\geq \frac{f}{q}.
\end{aligned}
$$

Assuming light is uniformly distributed over the circle of radius x'_q, the percentage of light due to mode q hitting the receiver at a distance z is

$$\frac{a_q}{\pi(x'_q)^2} \cdot A_{\text{receiver}}. \tag{21}$$

2.4 Gaussian Distribution Power Calculations

Because lasers are often used to illuminate diffractive optical elements, it is useful to incorporate gaussian beams into the analysis.

2.4.1 Complex Gaussian Parameter

Recall the complex gaussian parameter, $q(z)$ is defined as:

$$\frac{1}{q(z)} = \frac{1}{R(z)} - i\frac{\lambda}{n\pi w^2(z)}. \tag{22}$$

The gaussian field is then given by

$$E(x,y,z) = E_0\sqrt{\frac{2}{\pi}}\frac{1}{w(z)}\exp\left[-i[kz-\eta(z)] - i\frac{kr^2}{2q(z)}\right], \tag{23}$$

where

$$\eta(z) = \tan^{-1}\left[\frac{z}{\pi w_0^2 n}\right], \tag{24}$$

and $k = \frac{2\pi n}{\lambda}$. n is the refractive index in which the gaussian beam is propagating. Power is the integral of the intensity squared, and $I = EE^*$. E is normalized, using

$$\int_0^{2\pi}\int_0^{\infty}\exp[-2r^2/w^2]r\,dr\,d\theta = \frac{\pi}{2}w^2. \tag{25}$$

Since power is conserved, the total power in the beam is E_0^2. To be consistent with our notation, $a = E_0^2$.

In the complex gaussian parameter q, R is the radius of curvature, and w is the beam waist. The evolution of the complex gaussian parameter as it propagates through a sequence of elements (propagation, lenses, interfaces) can be followed by applying the ABCD law as follows.

$$q_2 = \frac{Aq_1 + B}{Cq_1 + D}, \tag{26}$$

where A, B, C, D are the elements of the ray matrix which describes the relation of a ray in plane 1 to a ray in plane 2. If there is a sequence of elements, then the total ray matrix is the matrix product of the individual element ray matrices. The order of the product is important, with the last element's matrix occurring at the left of the product.

The ray matrix for propagation of a distance d in a homogeneous medium is

$$\begin{vmatrix} 1 & d \\ 0 & 1 \end{vmatrix} \tag{27}$$

and the ray matrix for a lens of focal length f, ($f > 0$ converging, $f < 0$ diverging) is

$$\begin{vmatrix} 1 & 0 \\ -\frac{1}{f} & 1 \end{vmatrix} \tag{28}$$

For the case of a beam propagating through a lens with effective focal length $\frac{f}{q}$ and then propagating a distance d, the total ray matrix (hence the subscript $_T$) is

$$\begin{vmatrix} A_T & B_T \\ C_T & D_T \end{vmatrix} = \begin{vmatrix} 1 & d \\ 0 & 1 \end{vmatrix}\begin{vmatrix} 1 & 0 \\ -\frac{1}{\frac{f}{q}} & 1 \end{vmatrix} = \begin{vmatrix} 1-\frac{d\cdot q}{f} & d \\ -\frac{q}{f} & 1 \end{vmatrix}. \tag{29}$$

Let's assume the gaussian beam hitting the DOE is a plane wave, with waist, w_0. Then the complex gaussian parameter is

$$q_1 = \frac{in\pi w_0^2}{\lambda} \tag{30}$$

If the DOE is not at the laser waist, then q_1 could have a real component.

The complex gaussian parameter at a distance d away from the DOE, can be found by applying Equations 26 and 29. Then the waist can be calculated from Equation 22.

2.4.2 Power

Once the waist at the distance z is found, then the relative power hitting a receiver of radius $R_{receiver}$ centered on the optical axis can be calculated using

$$a_q \left[1 - \exp\left[\frac{-2R^2_{receiver}}{w^2(z)} \right] \right]. \tag{31}$$

To find the relative power hitting a more general location at a distance z from the DOE, the following two equations may be used:

$$\text{relative power} = a_q \frac{1}{2\pi} [\theta_{end} - \theta_{start}] \left[\exp\left[\frac{-2r^2_{inner}}{w^2(z)} \right] - \exp\left[\frac{-2r^2_{outer}}{w^2(z)} \right] \right], \tag{32}$$

and

$$\text{relative power} = \frac{2a_q}{\pi w^2(z)} \int_{x_1}^{x_2} \int_{y_1}^{y_2} \exp\left[\frac{-2(x^2+y^2)}{w^2(z)} \right] dxdy. \tag{33}$$

The first equation has the area of interest specified by polar coordinates, $r_{inner}, r_{outer}, \theta_{start}, \theta_{end}$. The second equation has the area of interest specified by x and y coordinates, x_1, x_2, y_1, y_2.

3 Implementation in Mathematica

The analysis presented in the previous sections has been implemented in Mathematica. The package DOEefficiency.m should be read in to set up the needed definitions and functions.

The variables and parameters for the specific problem to be analyzed can be read in from a file. An example of this is shown in the section Variables and Parameters. The parameters and variables can be changed at any time during a Mathematica session.

Finally, a Mathematica script called maketable.m is presented. Each time it is read in, it acts like a command to produce a table of pertinent results. The variables *NumModes* and *RecPlane* are used by maketable.m to determine the number of modes in the table, and the distance of the receiver plane. The output of maketable.m is saved in a file specified by the variable *outputfile*. This variable is a string, and can be changed at any time during a Mathematica session.

Maketable.m is just an example of how the package DOEefficiency.m can be used. Other scripts can be made to produce custom outputs.

3.1 Mathematica Package (DOEefficiency.m)

```
BeginPackage["DOEefficiency`"]

(* PARAMETERS FROM ACTUAL DOE SPECIFICATIONS
```

```
rReceiver = radius of receiver aperture
Diam = diameter of lens

(Only need 2 out of three of the following, since they are related:)
TO= period of grating in r squared
lambda0 = design wavelength
f0 = design focal length (T0 = 2 lambda0 f0)
(If you're comparing designs, then values for lambda0 and f0 are
known.  If you have an actual DOE, then you can use T0 (measured)
and lambda0, to calculate f0.)

muL = index of refraction in substrate
mu = index of refraction of grating
muR = index of refraction of media above grating

P = number of step levels in the  grating period
optimtheta0 is True if d1 is optimized for incident angle theta0;
    is False if d1 is optimized for normal incidence

waist0 = input waist
theta0 = angle of incidence for which d1 is optimized
lambda = wavelength, calculated as lambda0/lambdaratio
theta = angle of incidence upon grating
lambdaratio = ratio of lambda0 to lambda
*)


(* CALCULATED QUANTITIES (Intermediate):

fa = actual focal length

  thetaprime = angle inside grating
thetaL = angle in substrate

d1 = thickness of each step (can be optimized for normal
      or oblique incidence)

Phi1 = the phase delay from one step
*)

(* CALCULATED QUANTITIES (Final):
a[q] = relative power in order q
xprime[z,q] = radius of spot at distance z, due to order q
z=0 ==> location of the grating
ro[z,q] = ratio of area of order q to receiver
relpower[z,q] = relative percentage of light, due to order q,
 at dist z hitting recvr
cgp[z,q] = complex gaussian parameter at distance z, for order q
```

```
waist[z,q] = waist for gaussian beam at distance z, for order q
gaussrelpower[z,q] = relative power hitting the reciever, due
to order q, at distance z, assuming gaussian beams
relpowerpolarwindow[z,q,r1,r2,th1,th2] = the amount of relative
power in a window between radii r1 and r2, and
angles th1 and th2, at a distance z for order q,
assuming gaussian beams
relpowerwindow[z,q,x1,x2,y1,y2] = the amount of relative power in
a window between x1 and x2, and
y1 and y2, at a distance z for order q,
assuming gaussian beams
*)

lambda := lambda0/lambdaratio

fa::usage = "fa is the actual focal length of the DOE, with muL and muR
the surrounding indices of refraction"
faair := lambdaratio*f0
fa := faair*muL

thetaprime::usage = "thetaprime is the angle inside the DOE"
thetaprime := ArcSin[muR/mu * Sin[theta]]
theta0prime := ArcSin[muR/mu * Sin[theta0]]

(* theta0 = angle of incidence for which d1 is optimized *)
(*optimtheta0 is true if grating optimized for incident angle of theta *)

optimtheta0 = False
Phi1::usage = "Phi1 is the phase delay through one step level of the DOE"
Phi1 := 2*Pi*lambdaratio/P /; !optimtheta0 && theta==0
Phi1 := 2*Pi*lambdaratio/P *
(mu*Cos[theta] - muR*Cos[thetaprime])/
(mu*Cos[theta0] - muR*Cos[thetaprime0])*
(Cos[theta0] * Cos[thetaprime0])/(Cos[theta] * Cos[thetaprime])
/; optimtheta0
Phi1 := 2 * Pi * lambdaratio * (mu*Cos[theta] - muR*Cos[thetaprime])/
(P * Cos[theta] * Cos[thetaprime] * (mu - muR) )

a::usage = "a[order] gives the relative power in a
            given diffracted order"
a[orderval_]:=
N[Limit[
( Sin[Pi/P*(order - lambdaratio*muR*Sin[theta])]/
(Pi/P*(order - lambdaratio*muR*Sin[theta])) )^2 *
( Sin[P*(Phi1/2 -(order  - lambdaratio*muR*Sin[theta])*Pi/P)]/
(P*Sin[Phi1/2 -(order  - lambdaratio*muR*Sin[theta])*Pi/P]) )^2,
order->orderval]]
```

```
xprime::usage = "xprime is the radius of the
diffracted beam of a given order
at a given distance, assuming plane waves of uniform
unit amplitude. xprime is only meaningful for
distance greater than 0."
xprime[dist_,order_] := N[waist0*(order/fa*dist -1)] /; ((order>0)&&
(dist>= (fa/order)))
xprime[dist_,order_] := N[waist0*(1-dist*order/fa)] /; (order>0)&&
(dist<(fa/order)&&(dist>=0))
xprime[dist_,order_] := N[waist0*(-order/fa*dist +1)] /; (order<0) &&
(dist >=0)
xprime[dist_,order_] := N[waist0] /; order==0

ro::usage = "the fraction of power of a given order at a given distance
hitting the receiver of radius rReceiver, assuming plane waves of uniform
unit amplitude"
ro[dist_, order_] := 1/;  rReceiver > xprime[dist,order]
ro[dist_, order_] := 1/;  xprime[dist,order] == 0
ro[dist_, order_] := N[(rReceiver/xprime[dist,order])^2]

relpower::usage = "the relative power of a given
order at a given distance hitting the receiver,
                assuming plane waves of uniform unit amplitude"
relpower[dist_,order_] := ro[dist,order]*a[order]

cgp1 := (I*muR Pi waist0^2)/lambda
(* cpg1 could have a real part if DOE isn't at the laser waist *)

cgp::usage = "cgp is the complex Gaussian parameter of
the beam for a given order at a given distance"
cgp[dist_,order_] :=
(cgp1*(1-dist*order/fa) + dist)/
(cgp1*(-order/fa) + 1)

waist::usage = "waist is the gaussian beam waist for a given order at a
given distance. Waist is only meaningful for distance greater than 0."
waist[dist_,order_] := N[
Sqrt[lambda/
      (Im[1/cgp[dist,order]]*muL*Pi*(-1))
     ]
] /; dist >= 0

gaussrelpower::usage = "the relative power of a given order at a given
distance hitting the receiver, assuming Gaussian beams"
gaussrelpower[dist_,order_] := a[order]*
(1 - Exp[-2*rReceiver^2/(waist[dist,order]^2)])

relpowerpolarwindow::usage = "relpowerpolarwindow gives the
```

```
relative power in a window denoted by inner
  and outer radius, and theta start and theta end
for a given order and distance"
relpowerpolarwindow[dist_,order_,rin_,rout_,thstart_,thend_] :=
1/(2*Pi) *a[order]*(thend-thstart)*
(Exp[-2*rin^2/(waist[dist,order]^2)]-
Exp[-2*rout^2/(waist[dist,order]^2)])

relpowerwindow::usage = "relpowerwindow gives the relative power
in a window denoted by x1,x2, and y1, y2 for a given order and distance"
relpowerwindow[dist_,order_,x1_,x2_,y1_,y2_]:=
a[order]*
NIntegrate[2/(Pi*waist[dist,order]^2)*
Exp[-2*(x^2+ y^2)/(waist[dist,order]^2)],
{x,x1,x2},{y,y1,y2}]
diffeff[absorder_] :=
        Table[a[i],{i, -absorder, absorder}]

outputpwr[numorders_] :=
        Table[relpower[i*10^-6,j],{i,0,numorders},{j,1,numorders}]
EndPackage[]
```

3.2 Variables and Parameters (inputfile.m)

```
rReceiver = 100 10^-6
Diam = 100 10^-6
lambda0 = 980 10^-9
lambdaratio = 1
f0 = 100 10^-6
muL = 1.0
mu = 3.5
muR = 1.0
P = 4
theta = 0
waist0=30 10^-6
optimtheta0 = False
theta0 = 0
(*if optimtheta = True, then need to specify theta0*)

NumModes = 5 (* will study modes from - NumModes to + NumModes *)
RecPlane = fa (* the distance of the receiver plane *)

outputfile = "outputfile"
```

3.3 Mathematica Scripts (maketable.m)

```
defficiencies = diffeff[NumModes]
dradii = Table[xprime[RecPlane,i],{i,-NumModes,NumModes}]
dwaists = Table[waist[RecPlane,i],{i,-NumModes,NumModes}]
drpwr= Table[relpower[RecPlane,i],{i,-NumModes,NumModes}]
dgausspwr= Table[gaussrelpower[RecPlane,i],{i,-NumModes,NumModes}]

titles=Table[Mode[i],{i,-NumModes,NumModes}]
PutAppend[
OutputForm[
TableForm[{defficiencies,dradii,dwaists,drpwr,dgausspwr},
TableDirections->{Row,Column,Row,Column,Row},
TableSpacing->{3,1},
TableHeadings->{{"effic","radii","waists","rpwr","gausspwr"},titles}]]
,outputfile]
PutAppend["                                                            ",
outputfile]
PutAppend["P = ",P,outputfile]
PutAppend["Receiver plane at d = ",RecPlane,outputfile]
PutAppend["********************************************************** ",
outputfile]
PutAppend["                                                            ",
outputfile]
```

4 Examples

In this section we show some sample output using the files DOEefficiency.m, inputfile1.m and maketable.m. The files were generated using the following commands.

```
<<DOEefficiency.m
<<inputfile1.m

outputfile = "P2"
P = 2
<<maketable.m

outputfile = "P3"
P = 3
<<maketable.m

outputfile = "P4"
P = 4
<<maketable.m

outputfile = "P5"
P = 5
```

```
<<maketable.m

outputfile = "P6"
P = 6
<<maketable.m

outputfile = "P8"
P = 8
<<maketable.m

outputfile = "P8-8modes"
NumModes = 8
<<maketable.m

outputfile = "P4thetaPi36"
theta = Pi/36
P = 4
<<maketable.m

outputfile = "P4thetaPi72"
theta = Pi/72
NumModes = 5
<<maketable.m

outputfile = "recdistance"
theta = 0
RecPlane = 1.5 10^-4
<<maketable.m

Quit
```

The files generated are as follows.

4.1 Two Step Levels in Grating Period: "P2"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0.0162114	0.00018	0.000180003	0.00500352	0.00746666
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0.0450316	0.00012	0.000120005	0.031272	0.0338017
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0.405285	0.00006	0.000060009	0.405285	0.403715
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.405285	0.	$1.03981\ 10^{-6}$	0.405285	0.405285
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0.0450316	0.00006	0.000060009	0.0450316	0.0448573
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0.0162114	0.00012	0.000120005	0.0112579	0.0121686

```
"                                                      "
"p = "
2
"Receiver plane at d = "
0.0001
"****************************************************** "
"                                                      "
```

4.2 Three Step Levels in Grating Period: "P3"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0.0273567	0.00018	0.000180003	0.00844343	0.0126
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0	0.00012	0.000120005	0	0
Mode[-2]	0.170979	0.00009	0.000090006	0.170979	0.1565
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.683918	0.	$1.03981\ 10^{-6}$	0.683918	0.683918
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0.0427449	0.00009	0.000090006	0.0427449	0.039125
Mode[5]	0	0.00012	0.000120005	0	0

```
"                                                    "
"P = "
3
"Receiver plane at d = "
0.0001
"***************************************************** "
"                                                    "
```

4.3 Four Step Levels in Grating Period: "P4"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0	0.00018	0.000180003	0	0
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0.0900633	0.00012	0.000120005	0.0625439	0.0676035
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.810569	0.	$1.03981\ 10^{-6}$	0.810569	0.810569
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0.0324228	0.00012	0.000120005	0.0225158	0.0243372

```
"                                                          "
"P = "
4
"Receiver plane at d = "
0.0001
"********************************************************* "
"                                                          "
```

4.4 Five Step Levels in Grating Period: "P5"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0	0.00018	0.000180003	0	0
Mode[-4]	0.0546963	0.00015	0.000150004	0.0243095	0.032209
Mode[-3]	0	0.00012	0.000120005	0	0
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.87514	0.	$1.03981\ 10^{-6}$	0.87514	0.87514
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0	0.00012	0.000120005	0	0

```
"                                                         "
"P = "
5
"Receiver plane at d = "
0.0001
"******************************************************** "
"                                                         "
```

4.5 Six Step Levels in Grating Period: "P6"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0.0364756	0.00018	0.000180003	0.0112579	0.0168
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0	0.00012	0.000120005	0	0
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.911891	0.	$1.03981\ 10^{-6}$	0.911891	0.911891
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0	0.00012	0.000120005	0	0

```
"                                                        "
"p = "
6
"Receiver plane at d = "
0.0001
"********************************************************* "
"                                                        "
```

4.6 Eight Step Levels in Grating Period, 11 Modes Calculated: "P8"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0	0.00018	0.000180003	0	0
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0	0.00012	0.000120005	0	0
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.949641	0.	$1.03981\ 10^{-6}$	0.949641	0.949641
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0	0.00012	0.000120005	0	0

```
"                                                     "
"p = "
8
"Receiver plane at d = "
0.0001
"***************************************************** "
"                                                     "
```

4.7 Eight Step Levels in Grating Period, 17 Modes Calculated: "P8-8modes"

	effic	radii	waists	rpwr	gausspwr
Mode[-8]	0	0.00027	0.000270002	0	0
Mode[-7]	0.0193804	0.00024	0.000240002	0.00336466	0.00568519
Mode[-6]	0	0.00021	0.000210003	0	0
Mode[-5]	0	0.00018	0.000180003	0	0
Mode[-4]	0	0.00015	0.000150004	0	0
Mode[-3]	0	0.00012	0.000120005	0	0
Mode[-2]	0	0.00009	0.000090006	0	0
Mode[-1]	0	0.00006	0.000060009	0	0
Mode[0]	0	0.00003	0.000030018	0	0
Mode[1]	0.949641	0.	$1.03981\ 10^{-6}$	0.949641	0.949641
Mode[2]	0	0.00003	0.000030018	0	0
Mode[3]	0	0.00006	0.000060009	0	0
Mode[4]	0	0.00009	0.000090006	0	0
Mode[5]	0	0.00012	0.000120005	0	0
Mode[6]	0	0.00015	0.000150004	0	0
Mode[7]	0	0.00018	0.000180003	0	0
Mode[8]	0	0.00021	0.000210003	0	0

```
"                                                      "
"P = "
8
"Receiver plane at d = "
0.0001
"******************************************************* "
"                                                      "
```

4.8 Four Step Levels in Grating Period, Incident Angle $= \frac{\pi}{36}$: "P4thetaPi36"

	effic	radii	waists	rpwr	gausspwr
Mode[-8]	$9.1125\ 10^{-7}$	0.00027	0.000270002	$1.25\ 10^{-7}$	$2.18636\ 10^{-7}$
Mode[-7]	0.0136199	0.00024	0.000240002	0.00236456	0.00399535
Mode[-6]	0.000448798	0.00021	0.000210003	0.000101768	0.000163633
Mode[-5]	0.000159434	0.00018	0.000180003	0.0000492081	0.0000734324
Mode[-4]	$3.56769\ 10^{-6}$	0.00015	0.000150004	$1.58564\ 10^{-6}$	$2.1009\ 10^{-6}$
Mode[-3]	0.0717795	0.00012	0.000120005	0.0498469	0.0538793
Mode[-2]	0.00381742	0.00009	0.000090006	0.00381742	0.00349414
Mode[-1]	0.00349098	0.00006	0.000060009	0.00349098	0.00347747
Mode[0]	0.0078458	0.00003	0.000030018	0.0078458	0.0078458
Mode[1]	0.820964	0.	$1.03981\ 10^{-6}$	0.820964	0.820964
Mode[2]	0.00454486	0.00003	0.000030018	0.00454486	0.00454486
Mode[3]	0.000486292	0.00006	0.000060009	0.000486292	0.000484409
Mode[4]	$3.89264\ 10^{-6}$	0.00009	0.000090006	$3.89264\ 10^{-6}$	$3.56299\ 10^{-6}$
Mode[5]	0.0283434	0.00012	0.000120005	0.0196829	0.0212752
Mode[6]	0.000475649	0.00015	0.000150004	0.0002114	0.000280095
Mode[7]	0.0000863412	0.00018	0.000180003	0.0000266485	0.0000397671
Mode[8]	$9.5184\ 10^{-7}$	0.00021	0.000210003	$2.15837\ 10^{-7}$	$3.47043\ 10^{-7}$

```
"                                                     "
"P = "
4
"Receiver plane at d = "
0.0001
```

"** "

4.9 Four Step Levels in Grating Period, Incident Angle $= \frac{\pi}{72}$: "P4thetaPi72"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0.0000392628	0.00018	0.000180003	0.0000121182	0.0000180837
Mode[-4]	$2.50903\ 10^{-7}$	0.00015	0.000150004	$1.11512\ 10^{-7}$	$1.47749\ 10^{-7}$
Mode[-3]	0.0810384	0.00012	0.000120005	0.0562767	0.0608292
Mode[-2]	0.000958416	0.00009	0.000090006	0.000958416	0.000877252
Mode[-1]	0.000917026	0.00006	0.000060009	0.000917026	0.000913475
Mode[0]	0.00215619	0.00003	0.000030018	0.00215619	0.00215619
Mode[1]	0.820748	0.	$1.03981\ 10^{-6}$	0.820748	0.820748
Mode[2]	0.0010458	0.00003	0.000030018	0.0010458	0.0010458
Mode[3]	0.000114273	0.00006	0.000060009	0.000114273	0.000113831
Mode[4]	$2.62089\ 10^{-7}$	0.00009	0.000090006	$2.62089\ 10^{-7}$	$2.39894\ 10^{-7}$
Mode[5]	0.0305592	0.00012	0.000120005	0.0212217	0.0229384

```
"                                                       "
"p = "
4
"Receiver plane at d = "
0.0001
"******************************************************* "
"                                                       "
```

4.10 Four Step Levels in Grating Period, Receiver Distance = 150 microns: "recdistance"

	effic	radii	waists	rpwr	gausspwr
Mode[-5]	0	0.000255	0.000255005	0	0
Mode[-4]	0	0.00021	0.000210006	0	0
Mode[-3]	0.0900633	0.000165	0.000165007	0.0330811	0.0468581
Mode[-2]	0	0.00012	0.00012001	0	0
Mode[-1]	0	0.000075	0.0000750162	0	0
Mode[0]	0	0.00003	0.0000300405	0	0
Mode[1]	0.810569	0.000015	0.0000150809	0.810569	0.810569
Mode[2]	0	0.00006	0.0000600203	0	0
Mode[3]	0	0.000105	0.000105012	0	0
Mode[4]	0	0.00015	0.000150008	0	0
Mode[5]	0.0324228	0.000195	0.000195006	0.0085267	0.0132609

```
"                                                          "
"p = "
4
"Receiver plane at d = "
0.00015
"********************************************************* "
"                                                          "
```

5 Conclusion

We have presented the analysis of a general diffractive lens, addressing the issues of efficiency and power distribution of the various modes. The analysis described in the first part of this report is implemented using Mathematica. The package DOEefficiency.m contains the basic functions necessary to calculate a variety of quantities of interest in this problem. We have presented representative scripts which set values for the parameters of the problem and which can be used to produce output. Once DOEefficiency.m is read in, any Mathematica commands may be used, and other scripts can be written to produce custom output.

References

[1] Kai Chang, editor. *Handbook of Microwave and Optical Components*, volume 3. J. Wiley and Sons, 1990.

[2] H. Dammann. Blazed synthetic phase-only holograms. *Optik*, 31:95–104, 1970.

[3] H. Dammann. Spectral characteristics of stepped-phase gratings. *Optik*, 53(5):409–417, 1979.

[4] H. Nishihara and T. Suhara. Micro fresnel lenses. *Progress in Optics XXIV*, pages 1–37, 1987. Elsevier Science Publishers B.V.

[5] G. Swanson. Binary optics technology: The theory and design of multi-level diffractive optical elements. Technical Report TR 854, MIT Lincoln Labs, August 1989. Prepared for DARPA.

[6] G. Swanson. Binary optics technology: Theoretical limits on the diffraction efficiency of multilevel diffractive optical elements. Technical Report TR 914, MIT Lincoln Lab, March 1991. Prepared for DARPA.

[7] Stephen Wolfram. *Mathematica: A System for Doing Mathematics by Computer.* Addison-Wesley, 2nd edition, 1991.

[8] A. Yariv. *Optical Electronics.* Holt, Rinehart and Winston, 1985.

DISTRIBUTION:

1	MS 9018	Central Technical Files, 8523-2
5	MS 0899	Technical Library, 4414
1	MS 0619	Print Media, 12615
2	MS 0100	Document Processing, 7613-2
		For DOE/OSTI

Org.	Bldg.	Name	Rec'd by	Org.	Bldg.	Name	Rec'd by

Sandia National Laboratories